



Techniques of Water-Resources Investigations of the United States Geological Survey

Chapter B4

REGRESSION MODELING OF GROUND-WATER FLOW

By **Richard L. Cooley and Richard L. Naff**

Book 3
APPLICATIONS OF HYDRAULICS

6 Some Advanced Topics

6.1 Advanced Models

6.1.1 Regression When the Dependent Variable is Implicit

In all cases considered in previous sections, the assumption has been made that the deterministic part of the model equation can be solved explicitly for the dependent variable. However, this may not be true in some cases. An example of such a model written in terms of the true value for the dependent variable $f=f(\xi, \beta_1, \beta_2)$ is

$$\frac{\beta_1 k}{3(\beta_1 + \beta_2)} \left\{ \frac{1}{2} \ln \left(\frac{(k+f)^2}{k^2 - kf + f^2} \right) + \sqrt{3} \left[\tan^{-1} \left(\frac{2f-k}{k\sqrt{3}} \right) - \tan^{-1} \left(\frac{-1}{\sqrt{3}} \right) \right] \right\} - \xi = 0 \quad (6.1-1)$$

where $k = \sqrt[3]{\beta_1 + \beta_2}$ and $f=f(\xi, \beta_1, \beta_2)$ is the exact solution of equation 6.1-1. As can be seen, f is implicit in the model equation and cannot be directly solved for. A general deterministic form for an exact model (that is, a model that does not contain ϵ) where the dependent variable is implicit is

$$g[f(\xi, \beta), \xi, \beta] = 0 \quad (6.1-2)$$

where ξ and β are defined as usual.

Based on equation 6.1-2, a true regression model can be written in terms of observation vector \underline{Y} and disturbances $\underline{\epsilon}$ in the usual form

$$\underline{Y} = f(\underline{\xi}, \underline{\beta}) + \underline{\epsilon} \quad (6.1-3)$$

where $f(\underline{\xi}, \underline{\beta})$ is the vector of order n that is computed from

$$g[f(\underline{\xi}, \underline{\beta}), \underline{\xi}, \underline{\beta}] = 0 \quad (6.1-4)$$

In equation 6.1-4 vector g of order n represents n equations, each of which has the form of equation 6.1-2 written for an observation point. As an example, equation 6.1-1 would be written in the form

$$g_i = \frac{\beta_1 k}{3(\beta_1 + \beta_2)} \left\{ \frac{1}{2} \ln \left(\frac{(k+f_i)^2}{k^2 - kf_i + f_i^2} \right) + \sqrt{3} \left[\tan^{-1} \left(\frac{2f_i - k}{k\sqrt{3}} \right) - \tan^{-1} \left(\frac{-1}{\sqrt{3}} \right) \right] \right\} - \xi_i = 0, \quad i=1, 2, \dots, n \quad (6.1-5)$$

Note that in equation 6.1-5 only f evaluated at point i (that is, f_i) appears in the equation to compute g_i . However, in general this equation could contain values of f evaluated at any number of the possible points $j=1, 2, \dots, n$. An example of this type of model is the numerical model discussed in the next section.

The estimated regression model derived from equations 6.1-3 and 6.1-4 is

$$\underline{Y} = f(\underline{\xi}, \underline{b}) + \underline{e} \quad (6.1-6)$$

and

$$g[f(\underline{\xi}, \underline{b}), \underline{\xi}, \underline{b}] = 0 \quad (6.1-7)$$

where \underline{b} and \underline{e} are, as usual, general estimates of $\underline{\beta}$ and $\underline{\epsilon}$, respectively.

The general approach to solving the implicit-variable problem is very similar to that followed in section 3.3.1. First, the dependent variable values are written using a Taylor series expansion about an initial set of parameters. Then, from this, the linearized regression problem is set up and solved recursively to give the final solution to the nonlinear problem.

Taylor series expansion of f about an arbitrary initial set of parameters \underline{b}_0 can be written in the form of equation 3.3-5,

$$f(\underline{\xi}, \underline{b}) \approx f_0 + \underline{X}_0(\underline{b} - \underline{b}_0) \quad (6.1-8)$$

where

$$f_0 = f(\underline{\xi}, \underline{b}_0) \quad (6.1-9)$$

$$\underline{X}_0 = \left\{ \frac{\partial f_i}{\partial b_j} \bigg|_{\underline{b} = \underline{b}_0} \right\} \quad (6.1-10)$$

By using equation 6.1-6, equation 6.1-8 can be modified to give the estimated linearized regression model

$$Y - f_0 = \underline{X}_0(\underline{b} - \underline{b}_0) + e \quad (6.1-11)$$

which is exactly the same as the model used for the standard nonlinear case discussed in section 3.3.1.

To solve the linearized regression model based on equation 6.1-11, f_0 can be computed from equation 6.1-7 with $\underline{b} = \underline{b}_0$, and \underline{X}_0 can be computed by implicitly differentiating equation 6.1-7 with respect to b_j ($j=1,2,\dots,p$), setting $\underline{b} = \underline{b}_0$, then solving for \underline{X}_0 . To accomplish this computation of \underline{X}_0 , note that for any differential change $d\underline{b}$ in parameter vector \underline{b} to produce a new solution f of $\underline{g} = 0$, the total differential $d\underline{g}$ must equal zero because \underline{g} is always zero. Hence, by employing the chain rule of calculus, there results

$$d\underline{g} = \left(\frac{\partial \underline{g}}{\partial b_j} + \underline{M} \frac{\partial f}{\partial b_j} \right) db_j = 0, \quad j=1,2,\dots,p \quad (6.1-12)$$

where $\underline{M} = \{M_{ij}\} = \{\partial g_i / \partial f_j\}$. Note that if g_i contains only f_i , then \underline{M} is diagonal. Equation 6.1-12 can be evaluated using $\underline{b} = \underline{b}_0$ and $f_0 = f(\underline{\xi}, \underline{b}_0)$ to give

$$\left(\frac{\partial f}{\partial b_j} \right)_0 = \underline{X}_j^0 = -\underline{M}_0^{-1} \left(\frac{\partial \underline{g}}{\partial b_j} \right), \quad j=1,2,\dots,p \quad (6.1-13)$$

where

$$\underline{M}_0 = \left\{ \frac{\partial g_i}{\partial f_j} \right\}_{f=f_0} \quad (n \times n) \quad (6.1-14)$$

and subscript (or superscript) 0 means that the quantity is evaluated using $\underline{b} = \underline{b}_0$ and $f = f_0$. Solution of equation 6.1-7 for f_0 (using, for example, Newton iteration) followed by solution of equation 6.1-13 for \underline{X}_0 provides a convenient method of obtaining initial values f_0 and \underline{X}_0 from initial parameter estimates \underline{b}_0 . However, for subsequent iterations this method can be time consuming because it involves solving $\underline{g} = 0$ each time a new vector f and a new set of sensitivity vectors \underline{X}_j are to be computed from an updated parameter set.

A good method for computing good approximate values of f and \underline{X} corresponding to some arbitrary parameter set \underline{b} that is close to \underline{b}_0

involves approximating \underline{g} with another Taylor series expansion. If \underline{b} is close enough to \underline{b}_0 to allow dropping all terms except linear terms, then Taylor series expansion of equation 6.1-7 about initial set of dependent variable values f_0 can be written as

$$\begin{aligned} g[f(\underline{\xi}, \underline{b}), \underline{\xi}, \underline{b}] &= 0 \\ &\cong \underline{g}_0(\underline{b}) + \underline{M}_0(\underline{b})(f - f_0) \end{aligned} \quad (6.1-15)$$

where $\underline{g}_0(\underline{b}) = \underline{g}(f_0, \underline{\xi}, \underline{b})$, $\underline{M}_0(\underline{b}) = \{\partial g_i / \partial f_j |_{f=f_0}\}$ and $f \cong f(\underline{\xi}, \underline{b})$. By knowing \underline{b} , equation 6.1-15 may be solved for f . Corresponding values of \underline{X} are obtained as follows. If equation 6.1-15 is implicitly differentiated with respect to b_j ($j=1,2,\dots,p$), there results

$$\frac{\partial \underline{g}_0(\underline{b})}{\partial b_j} = -\underline{M}_0(\underline{b}) \frac{\partial f}{\partial b_j} - \frac{\partial \underline{M}_0(\underline{b})}{\partial b_j} (f - f_0) \quad (6.1-16)$$

or,

$$\frac{\partial f}{\partial b_j} = \underline{X}_j = -\underline{M}_0^{-1}(\underline{b}) \left(\frac{\partial \underline{g}_0(\underline{b})}{\partial b_j} + \frac{\partial \underline{M}_0(\underline{b})}{\partial b_j} (f - f_0) \right), \quad j=1,2,\dots,p \quad (6.1-17)$$

By using the above results, solution of the nonlinear regression problem is obtained by a procedure analogous to the procedure followed for the standard nonlinear problem. As indicated previously, to begin the first iteration assume an initial set of parameters \underline{b}_0 and solve equation 6.1-7 for $f_0 = f(\underline{\xi}, \underline{b}_0)$. Then solve equation 6.1-13 for \underline{X}_0 . Next, form and solve normal equation 3.3-10 by minimizing $S(\underline{b})$ (given by equation 3.3-4) with respect to \underline{b} , then scaling the resulting equations with \underline{C}_0 . That is, form and solve for $\underline{\delta}_1$

$$\underline{S}_0^T \omega \underline{S}_0 \underline{\delta}_1 = \underline{S}_0^T \omega (Y - f_0) \quad (6.1-18)$$

where

$$\underline{S}_0 = \underline{X}_0 \underline{C}_0 \quad (6.1-19)$$

$$\underline{\delta}_1 = \underline{C}_0^{-1} (\underline{b}_1 - \underline{b}_0). \quad (6.1-20)$$

For the second iteration, solve equation 6.1-15, written using $\underline{b}=\underline{b}_1$, for $\underline{f}-\underline{f}_0=\underline{f}_1-\underline{f}_0$. The result can be stated as

$$\underline{u}_1 = -\underline{M}_1^{-1} \underline{g}_1 \quad (6.1-21)$$

where subscript 1 on \underline{M} and \underline{g} indicates evaluation using the most recent values available for \underline{f} and \underline{b} (that is, \underline{f}_0 and \underline{b}_1), and

$$\underline{u}_1 = \underline{f}_1 - \underline{f}_0 \quad (6.1-22)$$

Next, solve equation 6.1-17 for \underline{X}_1 :

$$\underline{X}_j^1 = -\underline{M}_1^{-1} \left[\left(\frac{\partial \underline{g}_0(\underline{b})}{\partial b_j} \right)_1 + \left(\frac{\partial \underline{M}_0(\underline{b})}{\partial b_j} \right)_1 \underline{u}_1 \right], \quad (6.1-23)$$

$$j=1,2,\dots,p$$

where $(\cdot)_1$ indicates evaluation using $\underline{b}=\underline{b}_1$. Finally, form and solve the normal equations, written in terms of \underline{S}_1 and \underline{f}_1 , for $\underline{\delta}_2$.

For general iteration r , the equations to solve are

$$\underline{u}_r = -\underline{M}_r^{-1} \underline{g}_r \quad (6.1-24)$$

$$\underline{X}_j^r = -\underline{M}_r^{-1} \left[\left(\frac{\partial \underline{g}_{r-1}(\underline{b})}{\partial b_j} \right)_r + \left(\frac{\partial \underline{M}_{r-1}(\underline{b})}{\partial b_j} \right)_r \underline{u}_r \right], \quad (6.1-25)$$

$$j=1,2,\dots,p$$

$$\underline{f}_r = \underline{u}_r + \underline{f}_{r-1} \quad (6.1-26)$$

$$\underline{S}_r^T \underline{\omega} \underline{S}_r \underline{\delta}_{r+1} = \underline{S}_r^T \underline{\omega} (\underline{Y} - \underline{f}_r) \quad (6.1-27)$$

$$\underline{b}_{r+1} = \underline{C}_r \underline{\delta}_{r+1} + \underline{b}_r \quad (6.1-28)$$

where

$$\underline{S}_r = \underline{X}_r \underline{C}_r, \quad (6.1-29)$$

$$\underline{f}_0 = \underline{f}(\underline{\xi}, \underline{b}_0) \text{ so that } \underline{g}_0 = \underline{0}, \text{ and } \underline{f}_{-1} = \underline{f}_0.$$

At convergence of the solution \underline{g}_r , \underline{u}_r , and $\underline{\delta}_{r+1}$ all tend to zero so that $\underline{f}_r = \underline{f}(\underline{\xi}, \underline{\hat{b}})$ where $\underline{\hat{b}} = \underline{b}_{r+1} \cong \underline{b}_r$. At this point $S(\underline{\hat{b}}) = (\underline{Y} - \underline{f}(\underline{\xi}, \underline{\hat{b}}))^T \underline{\omega} (\underline{Y} - \underline{f}(\underline{\xi}, \underline{\hat{b}}))$ is at a minimum, and the nonlinear regression problem has been solved.

The solution procedure given by equations 6.1-24 through 6.1-29 can actually be considered to be a generalization of the Gauss-Newton procedure discussed in section 3.3.1, because if the standard nonlinear model is stated in the form

$$\underline{g} = \underline{f}(\underline{\xi}, \underline{b}) - \underline{f} = \underline{0} \quad (6.1-30)$$

then

$$\underline{g}_r = \underline{f}(\underline{\xi}, \underline{b}_r) - \underline{f}_{r-1} \quad (6.1-31)$$

$$\underline{M}_r = -\underline{I} \quad (6.1-32)$$

$$\left(\frac{\partial \underline{g}_{r-1}(\underline{b})}{\partial b_j} \right)_r = \left(\frac{\partial \underline{f}}{\partial b_j} \right)_r = \underline{X}_j^r \quad (6.1-33)$$

and equations 6.1-24 through 6.1-29 become the standard Gauss-Newton algorithm.

Iteration parameters ρ and μ should be applied to the present method in the same manner as they are for the Gauss-Newton method. Use of ρ to modify step size $\underline{\delta}_{r+1}$ leads to equation 3.3-19 ($\underline{b}_{r+1} = \rho \underline{C}_r \underline{\delta}_{r+1} + \underline{b}_r$) to compute \underline{b}_{r+1} . To employ μ , equation 6.1-27 is transformed to

$$(\underline{S}_r^T \underline{\omega} \underline{S}_r + \mu \underline{I}) \underline{\delta}_{r+1} = \underline{S}_r^T \underline{\omega} (\underline{Y} - \underline{f}_r). \quad (6.1-34)$$

The method for solution of the implicit-variable model given here requires the same three conditions to guarantee convergence to a global minimum as discussed for the modified Gauss-Newton method in section 3.3.3. However, in addition, the method requires that \underline{M} and $\partial \underline{g} / \partial b_j$ ($j=1,2,\dots,p$) be continuous and unique for all \underline{b} belonging to region R (see equation 3.3-25).

Solution Algorithm.

1. Before the first iteration, solve equation 6.1-7 for \underline{f}_0 using an initial estimate \underline{b}_0 for \underline{b} , and set $\underline{f}_{-1} = \underline{f}_0$.
2. Solve equation 6.1-24 and equation 6.1-25 for \underline{u}_r and \underline{X}_j^r ($j=1,2,\dots,p$).
3. Solve equation 6.1-26 for \underline{f}_r .
4. Solve equation 6.1-34 for $\underline{\delta}_{r+1}$.
5. Solve equation 3.3-17 for \underline{d}_{r+1} .
6. Solve equation 3.3-19 for \underline{b}_{r+1} .

7. Test to determine if $|d_i^{r+1}/c| > \epsilon$ ($i=1,2,\dots,p$). (See step 5, Gauss-Newton algorithm.)
8. If $|d_i^{r+1}/c| > \epsilon$, increment r by one and return to 2. If not, then the process has converged.

6.1.2 Regression When the Implicit-Variable Model is Numerical

If the numerical model assumes the general form of equation 3.3-21, which for convenience is restated here as

$$\underline{D}(h, \xi, \beta)h = q(h, \xi, \beta), \quad (6.1-35)$$

then the method derived in the previous section can be applied. The solution can be conceptually developed in two stages, first making the assumption that numerical solution points coincide exactly with observation points, which implies $m=n$, then, second, relaxing the assumption by following either of the two procedures described to obtain f from h for the Gauss-Newton method in section 3.3.2.

To develop the first stage of the solution, first note that because $m=n$, equation 6.1-35 may be written in the form $\underline{g}=\underline{0}$ analogous to equation 6.1-7:

$$\underline{g} = q - \underline{D}h = \underline{0} \quad (6.1-36)$$

where $h=f$. Next, expand equation 6.1-36 in a Taylor series to give equations exactly analogous to equations 6.1-24 and 6.1-25. Pertinent quantities in these equations are given by

$$\underline{M}_i^r = \left(\frac{\partial q}{\partial h_i} \right)_r - \left(\frac{\partial \underline{D}}{\partial h_i} \right)_r h_{r-1} - \underline{D}_i^r, \quad i=1,2,\dots,m \quad (6.1-37)$$

$$\left(\frac{\partial \underline{g}_{r-1}(b)}{\partial b_j} \right)_r = \left(\frac{\partial q}{\partial b_j} \right)_r - \left(\frac{\partial \underline{D}}{\partial b_j} \right)_r h_{r-1} \quad (6.1-38)$$

$$\left(\frac{\partial \underline{M}_i^{r-1}(b)}{\partial b_j} \right)_r = \left(\frac{\partial^2 q}{\partial b_j \partial h_i} \right)_r - \left(\frac{\partial^2 \underline{D}}{\partial b_j \partial h_i} \right)_r h_{r-1} - \left(\frac{\partial \underline{D}_i}{\partial b_j} \right)_r \quad (6.1-39)$$

where subscript i on a matrix denotes column i of the matrix. By using equations 6.1-38 and 6.1-39, an equation analogous to 6.1-25 can be written

$$\left(\frac{\partial h}{\partial b_j} \right)_r = -\underline{M}_r^{-1} \left\{ \left(\frac{\partial q}{\partial b_j} \right)_r - \left(\frac{\partial \underline{D}}{\partial b_j} \right)_r h_{r-1} + \sum_{i=1}^m \left[\left(\frac{\partial^2 q}{\partial b_j \partial h_i} \right)_r - \left(\frac{\partial^2 \underline{D}}{\partial b_j \partial h_i} \right)_r h_{r-1} - \left(\frac{\partial \underline{D}_i}{\partial b_j} \right)_r \right] \cdot u_i^r \right\}, \quad j=1,2,\dots,p \quad (6.1-40)$$

The second stage results from using one of the two procedures for obtaining f from h described in section 3.3.2 to obtain f from h and X_j from $\partial h / \partial b_j$ in the present case. With f and X defined, the solution algorithm of section 6.1.1 can be applied directly.

6.2 Modified Beale's Measure of Nonlinearity

Most of the methodology discussed here to analyze regression models is based on the assumption that the model is linear in the parameters. In the case that the model is nonlinear, Beale (1960, p. 54-55) developed an empirical measure of degree of nonlinearity with respect to the confidence regions on parameters. However, Guttman and Meeter (1965, p. 635) showed that if the degree of nonlinearity is high, Beale's measure seriously underestimates it. To correct for this underestimation problem, Linssen (1975) modified Beale's measure. More recently, Bates and Watts (1980) developed measures of nonlinearity based on the concepts of differential geometry. Although these measures are based on a much more extensive theory than Beale's (1960) measure or Linssen's (1975) modification, they also require extensive computation. Thus, here Beale's empirical measure as modified by Linssen is extended to give an approximate indication of the degree of average model nonlinearity.

To develop the measure, consider a linearized model of the form of equation 5.1-1, where, for convenience, general estimate b_i replaces β , and b_0 is set equal to \hat{b} to result in

$$\hat{f}_\ell^0 = \hat{f}_\ell + \underline{X}(b_\ell - \hat{b}) \quad (6.2-1)$$

where $\hat{f}_\ell = f(\xi, \hat{b})$. To emphasize that \hat{f}_ℓ^0 is a linearized estimate of $f(\xi, \hat{b}) = \hat{f}_\ell$, a superscript 0 is appended to \hat{f}_ℓ . If \hat{f}_ℓ^0 is calculated from equation 6.2-1 for m sets of parameter vectors b_ℓ ($\ell=1, 2, \dots, m$), then a measure c^2 of model nonlinearity in the region covered by the varied parameter sets is

$$c^2 = \sum_{\ell=1}^m (\hat{f}_\ell - \hat{f}_\ell^0)^T \underline{\omega} (\hat{f}_\ell - \hat{f}_\ell^0) \quad (6.2-2)$$

Equation 6.2-2 is the sum of squared distances (that is, squared lengths of vectors) between points $\underline{\omega}^{1/2} \hat{f}_\ell^0$ and $\underline{\omega}^{1/2} \hat{f}_\ell$ in observation space. (Recall that the distance between two points is the length of the vector joining the points, and that the squared length of a vector is given by the sum of squared lengths of its components.)

As explained further on, the most useful measure of nonlinearity is obtained by multiplying equation 6.2-2 by the quantity

$$qs^2 / \sum_{\ell=1}^m [(\hat{f}_\ell^0 - \hat{f}_\ell)^T \underline{\omega} (\hat{f}_\ell^0 - \hat{f}_\ell)]^2$$

to obtain

$$\hat{N}_b = qs^2 \frac{\sum_{\ell=1}^m (\hat{f}_\ell - \hat{f}_\ell^0)^T \underline{\omega} (\hat{f}_\ell - \hat{f}_\ell^0)}{\sum_{\ell=1}^m [(\hat{f}_\ell^0 - \hat{f}_\ell)^T \underline{\omega} (\hat{f}_\ell^0 - \hat{f}_\ell)]^2} \quad (6.2-3)$$

which is an extension for $q \leq p$ of Linssen's (1975) modification of Beale's measure of nonlinearity (Beale, 1960, p. 54-55).

Equation 6.2-3 can be justified as follows (see also Guttman and Meeter, 1965, and Linssen, 1975). The weighted distance between \hat{f}_ℓ^0 and \hat{f}_ℓ is designated ϵd so that the geometric relationships among weighted vectors \hat{f}_ℓ , \hat{f}_ℓ^0 , and $\hat{f}_\ell^0 - \hat{f}_\ell$ can be diagrammed in observation space as shown in figure 6.2-1. Now,

$$\begin{aligned} \epsilon^2 &= \frac{(\epsilon d)^2}{d^2} = \frac{d^2 (\epsilon d)^2}{(d^2)^2} \\ &= d^2 \frac{(\hat{f}_\ell - \hat{f}_\ell^0)^T \underline{\omega} (\hat{f}_\ell - \hat{f}_\ell^0)}{[(\hat{f}_\ell^0 - \hat{f}_\ell)^T \underline{\omega} (\hat{f}_\ell^0 - \hat{f}_\ell)]^2} \end{aligned} \quad (6.2-4)$$




Figure 6.2-1

where, by definition,

$$\begin{aligned} d^2 &= (\hat{f}_\ell^0 - \hat{f}_\ell)^T \underline{\omega} (\hat{f}_\ell^0 - \hat{f}_\ell) \\ &= [\underline{X}(b_\ell - \hat{b})]^T \underline{\omega} [\underline{X}(b_\ell - \hat{b})] \\ &= (b_\ell - \hat{b})^T \underline{X}^T \underline{\omega} \underline{X} (b_\ell - \hat{b}) \end{aligned} \quad (6.2-5)$$

To obtain a more convenient form for d^2 , note that

$$\begin{aligned} &(\underline{Y} - \hat{f}_\ell^0)^T \underline{\omega} (\underline{Y} - \hat{f}_\ell^0) - (\underline{Y} - \hat{f}_\ell)^T \underline{\omega} (\underline{Y} - \hat{f}_\ell) \\ &= (\underline{Y} - \hat{f}_\ell - \underline{X}(b_\ell - \hat{b}))^T \underline{\omega} (\underline{Y} - \hat{f}_\ell - \underline{X}(b_\ell - \hat{b})) \\ &\quad - (\underline{Y} - \hat{f}_\ell)^T \underline{\omega} (\underline{Y} - \hat{f}_\ell) \\ &= -2(b_\ell - \hat{b})^T \underline{X}^T \underline{\omega} (\underline{Y} - \hat{f}_\ell) + (b_\ell - \hat{b})^T \underline{X}^T \underline{\omega} \underline{X} (b_\ell - \hat{b}) \\ &= (b_\ell - \hat{b})^T \underline{X}^T \underline{\omega} \underline{X} (b_\ell - \hat{b}) \end{aligned} \quad (6.2-6)$$

where equation 5.1-9 was used. The combination of equations 6.2-5 and 6.2-6 shows that

$$d^2 = (\underline{Y} - \hat{f}_\ell^0)^T \underline{\omega} (\underline{Y} - \hat{f}_\ell^0) - (\underline{Y} - \hat{f}_\ell)^T \underline{\omega} (\underline{Y} - \hat{f}_\ell) \quad (6.2-7)$$

If \hat{f}_ℓ^0 is assumed to lie on the edge of the confidence region given by equation 5.6-12 so that $b_\ell^T = [\hat{b}_1^T \quad \hat{b}_2^T]$, then from equations 5.6-12 and 6.2-7 it can be seen that

$$d^2 = qs^2 F_\alpha(q, n-p) \quad (6.2-8)$$

Hence, if both numerator and denominator of equation 6.2-4 are averaged over m sets of parameters, there results

$$E(\epsilon^2) \cong \hat{N}_b F_\alpha(q, n-p) \quad (6.2-9)$$

Based on equation 6.2-9, Beale (1960, p. 60) ranked the degrees of nonlinearity as follows: The model is highly nonlinear if

$$\hat{N}_b > 1/F_\alpha(q, n-p) \quad (6.2-10)$$

because in this case $E(\epsilon^2) > 1$, and the discrepancy is actually greater than d . If

$$\hat{N}_b < 0.01/F_\alpha(q, n-p), \quad (6.2-11)$$

then the model is classed as being effectively linear because $E(\epsilon^2) < 0.01$. For points in between, Beale (1960, p. 60) stated that the linear theory is adequate to give a rough idea of significance but may not bring out full implications of the analysis. However, Guttman and Meeter (1965, p. 636) noted that equation 6.2-11 may be overly conservative to define a maximum value of \hat{N}_b for an approximately linear model. Experiments conducted by the authors indicate that, if

$$\hat{N}_b < 0.09/F_\alpha(q, n-p), \quad (6.2-12)$$

then confidence intervals given by linear theory are fairly good approximations of the exact ones as given by Vecchia and Cooley (1987). Thus, equation 6.2-12 is used to define the maximum value of \hat{N}_b to consider the model to be roughly linear.

Because equation 6.2-9 is justified by assuming that the points f_l^o lie on the edge of the confidence region, a reasonable way to obtain the points is to choose them from equation 5.6-14, although, as noted by Beale (1960, p. 55), the points do not have to lie on the edge of the confidence region. Thus, one could use $m \leq 2q$ sets of parameters $\underline{b}_l^T = [\underline{b}_{1l}^T, \underline{\beta}_2^T]$. Note that whether or not the model is linear, \underline{b}_{1l} and $\underline{\beta}_2$ corresponding to the partition of \underline{b} given in equation 5.6-3 are properly chosen without the necessity of performing additional least squares solutions to obtain each set \underline{b}_{1l} . This fact is true because subset \underline{b}_{1l} is required to lie on the edge of the linearized confidence region.

Rigorous use of equations 6.2-10 through 6.2-12 theoretically requires that disturbances be distributed normally. However, it would be convenient to be able to gauge the degree of nonlinearity of the model irrespective of the

properties of ϵ . If the confidence region in equation 5.6-12 were large enough to encompass virtually all physically plausible sets of parameters, then model nonlinearity as assessed using equations 6.2-10 through 6.2-12 would be meaningful. Based upon past experience, F values generated using $\alpha=0.05$ have been found to yield such a confidence region and thus to be adequate to gauge nonlinearity.

Problem 6.2-1

Four sets of parameters that correspond to four points on the edge of the linearized confidence region in equation 5.6-13 result from problem 5.6-2. These four sets of parameters can be subdivided into two groups of two. Pick two different parameter sets from the two groups and compute two corresponding sets of drawdowns at the observation points using the nonlinear (Theis) model. Then, using the modified Beale's measure program (appendix 6.4.1), find the modified Beale's measure. Is the model nearly linear?

Problem 6.2-2

Use the four parameter sets resulting from problem 5.6-3 in the nonlinear regression flow program of appendix 4.3.4, as augmented by the inserts of appendix 6.4.1, to compute the modified Beale's measure. Are the various statistical measures obtained from the linearized model approximately valid (at least as determined from the four parameter sets employed)?

6.3 Compatibility of Prior and Regression Estimates of Parameters

If the regression model contains prior information on the parameters, an important part of the analysis to determine whether or not the model is correct is to test the null hypothesis that the prior and sample information are in agreement; in other words,

$$H_0: E(\underline{Y}_p - \underline{f}_p(\underline{\xi}, \underline{b}_0)) = \underline{X}_p(\underline{\beta} - \underline{b}_0) \text{ versus}$$

$$H_1: E(\underline{Y}_p - \underline{f}_p(\underline{\xi}, \underline{b}_0)) \neq \underline{X}_p(\underline{\beta} - \underline{b}_0).$$

As indicated in section 5.5, graphical analysis of residuals can usually detect an incompatibility between sample and prior information. However, in some cases an additional test might be desired. Theil (1963) showed that the test statistic

$$\Gamma = (\underline{Y}_p - \underline{f}_p(\underline{\xi}, \underline{b}_0) - \underline{X}_p(\hat{\underline{b}}^* - \underline{b}_0))^T \cdot [\sigma^2 \underline{X}_p (\underline{X}_s^T \underline{\omega}_s \underline{X}_s)^{-1} \underline{X}_p^T + \sigma^2 \underline{\omega}_p^{-1}]^{-1} \cdot (\underline{Y}_p - \underline{f}_p(\underline{\xi}, \underline{b}_0) - \underline{X}_p(\hat{\underline{b}}^* - \underline{b}_0)) \quad (6.3-1)$$

where vector $\hat{\underline{b}}^*$ is the ordinary least squares estimate of vector $\underline{\beta}$, is Chi square distributed with n_p degrees of freedom ($\chi^2(n_p)$) provided that all of the assumptions given by equations 5.2-1 through 5.2-3, 5.2-6, and 5.2-10 hold true, σ^2 is known, and $\underline{\omega}$ is of the form

$$\underline{\omega} = \begin{bmatrix} \underline{\omega}_s & \underline{0} \\ \underline{0} & \underline{\omega}_p \end{bmatrix} \quad (6.3-2)$$

where $\underline{\omega}_s$ and $\underline{\omega}_p$ are known and symmetric positive definite of order n_s and n_p , respectively. If $\underline{\omega}$ is of the form

$$\underline{\omega} = \begin{bmatrix} \underline{V}_s^{-1} & \underline{0} \\ \underline{0} & \underline{U}^{-1} \sigma^2 \end{bmatrix} \quad (6.3-3)$$

and σ^2 is unknown, then the test statistic

$$\hat{\Gamma} = (\underline{Y}_p - \underline{f}_p(\underline{\xi}, \underline{b}_0) - \underline{X}_p(\hat{\underline{b}}^* - \underline{b}_0))^T \cdot [s^2 \underline{X}_p (\underline{X}_s^T \underline{V}_s^{-1} \underline{X}_s)^{-1} \underline{X}_p^T + \underline{U}]^{-1} \cdot (\underline{Y}_p - \underline{f}_p(\underline{\xi}, \underline{b}_0) - \underline{X}_p(\hat{\underline{b}}^* - \underline{b}_0)) \quad (6.3-4)$$

is asymptotically $\chi^2(n_p)$ distributed. If the computed value of $\hat{\Gamma}$, γ , is greater than $\chi_{\alpha}^2(n_p)$, where α indicates significance level, then the null hypothesis is rejected.

Problem 6.3-1

Using equation 6.3-4, test the compatibility

of the prior estimate of the boundary head, f_{p2} and the pure regression estimate, \hat{b}_2^* , of problem 3.2-1. To conduct this test you will have to do an ordinary least-squares regression. The model of appendix 4.3.4 may be employed for this in the same manner as for problem 4.2-1. Use the model output to obtain the necessary quantities in equation 6.3-4.

6.4 Appendix

6.4.1 Documentation of Program to Compute the Modified Beale's Measure

This program performs a straightforward computation of the modified Beale's measure, equation 6.2-3. Vectors \underline{f}_p , \underline{f}_p^0 , and \underline{f} are assumed to be composed of sample information and direct prior information on some or all parameters. The weight matrix for sample and prior information is assumed to be given in the form of equation 3.4-12, and the sensitivity matrix for the prior information is assumed to be given by equation 4.1-6.

There are two versions of the program. One is for general use, and all variables needed for the calculation must be read in. The other version is designed to be an integral part of the regression ground-water model documented in appendix 4.3.4 and requires only q and the extra sets of parameters needed for the Beale's measure calculation as input in addition to input already required for the regression solution.

The programs were developed using the Microsoft Fortran Compiler, Version 3.3, with the DOS 2.0 operating system on an IBM PC/XT computer with the IBM 8088 Math Coprocessor and 256 KB memory. Except for the OPEN statements near the beginning of the general code, Fortran 66 was used throughout to make the codes as machine independent as possible. The general source code is contained in file BEALE.FOR, and the version designed to be inserted into the regression code is contained in file BLEINS.FOR, both of which are in the diskette accompanying this report.

Input data for General Version.—Data Set A.
 Problem size information; one card (format 4I5, F10.0).

Line Columns	Variable	Definition
1-5	NVAR	Number of parameters, p .
6-10	NRES	Number of restrictions, q .
11-15	NOBS	Number of sample observations, n .
16-20	NPRIR	Number of regression parameters having direct prior information, n_p .
21-25	NPTS	Number of data sets to compute the modified Beale's measure, m .
26-35	VAR	Error variance, s^2 .

Data Set B.
 Estimated regression parameters, \hat{b} (format 8F10.0).

Line columns	Variable	Definition
1-10	BOPT(1)	Estimated regression parameters, entered sequentially from 1 through NVAR.
11-20	BOPT(2)	
⋮	⋮	
	BOPT(NVAR)	

Data Set C.
 Dependent variable vector for sample information, \hat{f}_s , computed using \hat{b} (format 8F10.0).

Line columns	Variable	Definition
1-10	FOPT(1)	Computed dependent variable values, entered sequentially from 1 through NOBS.
11-20	FOPT(2)	
⋮	⋮	
	FOPT(NOBS)	

Data Set D.
 Weight matrix for sample information, V_s^{-1} (format 8F10.0).

Line columns	Variable	Definition
1-10	W(1)	Diagonal weight matrix for sample information, entered sequentially from 1 through NOBS.
11-20	W(2)	
⋮	⋮	
	W(NOBS)	

Data Set E.
 Sensitivity matrix for sample information, X_s (format 8F10.0)

Line columns	Variable	Definition
1-10	X(1,1)	Sensitivity matrix for sample information, entered sequentially 1 through NVAR for each observation. Each new observation begins a new line, for a total of NOBS observations.
11-20	X(2,1)	
⋮	⋮	
	X(NVAR,1)	
1-10	X(1,2)	
⋮	⋮	
	X(NVAR,2)	
	X(NVAR,NOBS)	

Data Set F.

Parameter numbers having prior information (format 16I5).

Line columns	Variable	Definition
1-5	IPR(1)	Array subscript numbers for regression parameters in BOPT(1) having prior information, entered in any order from 1 through NPRIR.
6-10	IPR(2)	
⋮	⋮	
	IPR(NPRIR)	

Omit data set of NPRIR=0.**Data Set G.**Standard deviation matrix for prior information, $\underline{U}^{1/2}$ (format 8F10.0).

Line columns	Variable	Definition
1-10	WP(1)	Diagonal standard deviation matrix for prior information, entered in the same order as IPR(I) from 1 through NPRIR.
⋮	⋮	
	WP(NPRIR)	

Omit data set if NPRIR=0.**Data Set H.**Alternate parameter sets, b_i (format 8F10.0). This data set and the next one are read in sequence (H, I, H, I, ...) a total of NPTS times.

Line columns	Variable	Definition
1-10	B(1)	Alternate parameter sets, entered sequentially 1 through NVAR. Order must be the same as for BOPT(1).
11-20	B(2)	
⋮	⋮	
	B(NVAR)	

Data Set I.Alternate dependent variable vectors for sample information, f_{st} , computed using b_i (format 8F10.0).

Line columns	Variable	Definition
1-10	FC(1)	Alternate sample dependent variable values computed using the nonlinear model, entered sequentially 1 through NOBS. Order must be the same as for FOPT(1).
11-20	FC(2)	
⋮	⋮	
	FC(NOBS)	

Output for General Version.—Output is all clearly labeled; it is ordered as follows:

1. Data sets A through G.
2. Data sets H and I. Data for numbers 2 through 4 below are printed sequentially for each data set ℓ ($\ell=1,2,\dots,m$).
3. Dependent variable vector, $f_{s\ell}^o$, for sample information, computed using the linearized model.
4. Total sums of squared differences $(f_{\ell}-\hat{f})^T \cdot \underline{\omega}(f_{\ell}-\hat{f})$ and $(f_{\ell}^o-\hat{f})^T \underline{\omega}(f_{\ell}^o-\hat{f})$, where

$$\hat{f}_{\ell}^T = [f_{s\ell}^T \ f_{p\ell}^T], \quad f_{\ell}^{oT} = [f_{s\ell}^{oT} \ f_{p\ell}^T], \quad \hat{f}^T = [f_s^T \ f_p^T], \quad \text{and}$$

$$\underline{\omega} = \begin{bmatrix} \underline{V}_s^{-1} & \underline{0} \\ \underline{0} & \underline{U}^{-1}s^2 \end{bmatrix}.$$

5. Beale's measure, $\hat{N}_b = BN$.

Use of Version Integral with the Regression Ground-Water Program.—This version consists of sets of statements to be inserted into the program of appendix 4.3.4, as indicated on the appended listing. Input is the same as if a regression solution were to be obtained, except that the initial set of parameters must be the optimum set \hat{b} , and extra data relating to the modified Beale's measure is required. After entering data set T, use data sets U and V to enter the data for $\ell=1,2,\dots,m$ alternate solutions. Follow these data with a final line to input q and s^2 with format I5,F10.0. A complete regression solution is not obtained; only computations through the calculation of sensitivities on the first iteration are completed before proceeding to calculate the modified Beale's measure. Thus, output consists of regression output through number 19 (see "Output" in appendix 4.3.4) plus output analogous to numbers 2 through 4 of the general version of the modified Beale's measure program.

Program Listing for General Version.

```

C      MODIFIED BEALE'S MEASURE PROGRAM BY R. L. COOLEY, USGS, DENVER,
C      COLO.
      DIMENSION BOPT(20),FOPT(70),B(20),FC(70),FL(70),X(20,70)
      1,W(70),IPR(20),WP(20)
      COMMON/ITP/IIN,IOUT
      COMMON/FLT/X
      OPEN (5,FILE='BEALE.DAT',STATUS='OLD',ACCESS='SEQUENTIAL'
      1,FORM='FORMATTED')
      OPEN (6,FILE='BEALE.OUT',STATUS='NEW',ACCESS='SEQUENTIAL'
      1,FORM='FORMATTED')
C**FORMAT LIST
  1 FORMAT (5I5,F10.0)
  2 FORMAT (8F10.0)
  3 FORMAT (9H1NVAR = ,I4/9H NRES = ,I4/9H NOBS = ,I4
  1/9H NPRIR = ,I4/9H NPTS = ,I4/9H VAR = ,G11.5)
  4 FORMAT (1H0,26X,18HOPTIMUM PARAMETERS
  1/1H ,3X,3(3HNO.,9X,4HBOPT,8X))
  5 FORMAT (1H0,9X,52HDEPENDENT VARIABLES COMPUTED WITH OPTIMUM PARAME
  1TERS/1H ,3X,3(3HNO.,9X,4HFOPT,8X))
  6 FORMAT (1H0,21X,26HPARAMETERS FOR SAMPLE NO. ,I3
  1/1H ,3X,3(3HNO.,11X,1HB,9X))
  7 FORMAT (1H0,12X,44HDEPENDENT VARIABLES COMPUTED FOR SAMPLE NO. ,I3
  1/1H ,3X,3(3HNO.,10X,2HFC,9X))
  8 FORMAT (38H0 SENSITIVITIES FOR OPTIMUM PARAMETERS)
  9 FORMAT (1H0,6X,55HLINEARIZED DEPENDENT VARIABLES COMPUTED FOR SAMP
  1LE NO. ,I3/1H ,3X,3(3HNO.,10X,2HFL,9X))
 10 FORMAT (1H0,5HBN = ,G11.5)
 11 FORMAT (23HOSS((FC-FOPT)*W**.5) = ,G11.5
  1/23H SS((FL-FOPT)*W**.5) = ,G11.5)
 12 FORMAT (1H0,14X,42HRELIABILITY WEIGHTS FOR SAMPLE INFORMATION
  1/1H ,3X,3(3HNO.,10X,1HW,10X))
 13 FORMAT (16I5)
 14 FORMAT (1H0,12X,43HNO.S OF PARAMETERS HAVING PRIOR INFORMATION
  1/1H ,3X,3(3HNO.,8X,3HIPR,10X))
 15 FORMAT (1H0,14X,40HSTANDARD DEVIATIONS OF PRIOR INFORMATION
  1/1H ,3X,3(3HNO.,10X,2HWP,9X))
 16 FORMAT (6HOEV = ,G11.5)
C**DEFINE INPUT FILE, OUTPUT FILE, AND ARRAY DIMENSION
      IIN=5
      IOUT=6
      NVD=20
C**READ BASE DATA
      READ(IIN,1) NVAR,NRES,NOBS,NPRIR,NPTS,VAR
      WRITE(IOUT,3) NVAR,NRES,NOBS,NPRIR,NPTS,VAR
      READ(IIN,2) (BOPT(J),J=1,NVAR)
      WRITE(IOUT,4)
      CALL PRTOTB(BOPT,NVAR)
      READ(IIN,2) (FOPT(I),I=1,NOBS)
      WRITE(IOUT,5)
      CALL PRTOTB(FOPT,NOBS)
      READ(IIN,2) (W(I),I=1,NOBS)
      WRITE(IOUT,12)
      CALL PRTOTB(W,NOBS)

```

SET A

SET B

SET C

SET D

Program Listing for General Version—Continued

```

      DO 20 J=1,NOBS
      READ(IIN,2) (X(I,J),I=1,NVAR)
20  CONTINUE
      WRITE(IOUT,8)
      CALL PRTOT(X,NVAR,NOBS,NVD)
      IF(NPRIR.LT.1) GO TO 45
      READ(IIN,2) EV
      WRITE(IOUT,16) EV
      READ(IIN,13) (IPR(I),I=1,NPRIR)
      WRITE(IOUT,14)
      CALL PRTOTC(IPR,NPRIR)
      READ(IIN,2) (WP(I),I=1,NPRIR)
      WRITE(IOUT,15)
      CALL PRTOTB(WP,NPRIR)
      DO 40 I=1,NPRIR
40  WP(I)=EV/(WP(I)*WP(I))
C**READ DATA FOR EACH SAMPLE AND COMPUTE MODIFIED BEALE'S MEASURE, BN
45  SUMA=0.
      SUMB=0.
      DO 80 M=1,NPTS
      READ(IIN,2) (B(J),J=1,NVAR)
      WRITE(IOUT,6) M
      CALL PRTOTB(B,NVAR)
      READ(IIN,2) (FC(I),I=1,NOBS)
      WRITE(IOUT,7) M
      CALL PRTOTB(FC,NOBS)
      SUMC=0.
      SUMD=0.
      DO 60 J=1,NOBS
      SUM=FOPT(J)
      DO 50 I=1,NVAR
50  SUM=SUM+X(I,J)*(B(I)-BOPT(I))
      FL(J)=SUM
      TMP=FC(J)-SUM
      SUMA=SUMA+TMP*W(J)*TMP
      TMP=FC(J)-FOPT(J)
      SUMC=SUMC+TMP*W(J)*TMP
      TMP=SUM-FOPT(J)
      SUMD=SUMD+TMP*W(J)*TMP
60  CONTINUE
      IF(NPRIR.LT.1) GO TO 75
      DO 70 J=1,NPRIR
      I=IPR(J)
      TMP=B(I)-BOPT(I)
      TMP=TMP*WP(J)*TMP
      SUMC=SUMC+TMP
70  SUMD=SUMD+TMP
75  WRITE(IOUT,9) M
      CALL PRTOTB(FL,NOBS)
      WRITE(IOUT,11) SUMC,SUMD
80  SUMB=SUMB+SUMD*SUMD
      TMP=NRES
      BN=TMP*VAR*SUMA/SUMB

```

Program Listing for General Version—Continued

```

        WRITE(IOUT,10) BN
        STOP
        END
        SUBROUTINE PRTOTB(VAL,NO)
C**PRINT VALUES IN THREE GROUPS OF TWO COLUMNS
        DIMENSION VAL(NO)
        COMMON/ITP/IIN,IOUT
        NR=NO/3
        IF(3*NR.NE.NO) NR=NR+1
        DO 10 K=1,NR
        WRITE(IOUT,20) (L,VAL(L),L=K,NO,NR)
10 CONTINUE
        RETURN
20 FORMAT (1H ,2X,3(I3,7X,G11.5,3X))
        END
        SUBROUTINE PRTOTC(IVAL,NO)
C**PRINT INTEGERS IN THREE GROUPS OF TWO COLUMNS
        DIMENSION IVAL(NO)
        COMMON/ITP/IIN,IOUT
        NR=NO/3
        IF(3*NR.NE.NO) NR=NR+1
        DO 10 K=1,NR
        WRITE(IOUT,20) (L,IVAL(L),L=K,NO,NR)
10 CONTINUE
        RETURN
20 FORMAT (1H ,2X,3(I3,8X,I4,9X))
        END
        SUBROUTINE PRTOT(C,NR,NC,NRD)
C**PRINT MATRICES DIVIDED VERTICALLY INTO TEN-COLUMN BLOCKS
        DIMENSION C(NRD,NC)
        COMMON/ITP/IIN,IOUT
        DO 60 K=1,NC,10
        J10=K+9
        IF(J10.GT.NC) J10=NC
        WRITE(IOUT,70) (J,J=K,J10)
        WRITE(IOUT,90)
        DO 30 I=1,NR
30 WRITE(IOUT,80) I,(C(I,J),J=K,J10)
60 CONTINUE
70 FORMAT(1H0,10(9X,I3))
80 FORMAT (1H ,I3,1X,10(1X,G11.5))
90 FORMAT (1H )
        RETURN
        END

```

Listing of Inserts to the Regression Ground-Water Flow Program.

```

C
C**INSERT JUST BEFORE EQUIVALENCE STATEMENT FOR MODIFIED BEALE'S MEASURE
      DIMENSION BOPT(20),HOPT(70)
      EQUIVALENCE (P(1),BOPT(1)),(HO(1),HOPT(1))
C
C
C**INSERT AFTER STATEMENT LABEL 260 FOR MODIFIED BEALE'S MEASURE
      DO 1000 J=1,NVAR
1000  BOPT(J)=B(J)
      DO 1100 I=1,NOBS
      K=KOB(S(I)
1100  HOPT(I)=BK(I)*HC(K)+BL(I)*HC(K+1)+BM(I)*HC(K+ID)+BN(I)*HC(K+ID+1)
      SUMA=0.
      SUMB=0.
      GO TO 640
C
C
C**INSERT JUST BEFORE STATEMENT LABEL 690 FOR MODIFIED BEALE'S MEASURE
      SUMC=0.
      SUMD=0.
      WRITE(IOUT,2000)
2000  FORMAT (1H0,3X,28HCOMPUTED AND LINEARIZED HEADS/1H ,3X,3HNO.,7X
1,2HHC,13X,2HHL)
      DO 2200 J=1,NOBS
      K=KOB(S(J)
      HCJ=BK(J)*HC(K)+BL(J)*HC(K+1)+BM(J)*HC(K+ID)+BN(J)*HC(K+ID+1)
      HL=HOPT(J)
      DO 2100 I=1,NVAR
2100  HL=HL+X(I,J)*(B(I)-BOPT(I))
      TMP=HCJ-HL
      SUMA=SUMA+TMP*W(J)*TMP
      TMP=HCJ-HOPT(J)
      SUMC=SUMC+TMP*W(J)*TMP
      TMP=HL-HOPT(J)
      SUMD=SUMD+TMP*W(J)*TMP
      WRITE(IOUT,856) J,HCJ,HL
2200  CONTINUE
      IF(NPRIR.LT.1) GO TO 2400
      DO 2300 J=1,NVAR
      IF(WP(J).LT.1.E-10) GO TO 2300
      TMP=B(J)-BOPT(J)
      TMP=TMP*WP(J)*TMP
      SUMC=SUMC+TMP
      SUMD=SUMD+TMP
2300  CONTINUE
2400  WRITE(IOUT,2500) SUMC,SUMD
2500  FORMAT (23HOSS((HC-HOPT)*W**.5) = ,G11.5
1/23H SS((HL-HOPT)*W**.5) = ,G11.5)
      SUMB=SUMB+SUMD*SUMD
C
C
C**INSERT AFTER STATEMENT LABEL 690 FOR MODIFIED BEALE'S MEASURE

```

Listing of Inserts to the Regression Ground-Water Flow Program—Continued

```

      READ(IIN,812) NRES,VAR
      WRITE(IOUT,2600) NRES,VAR
2600  FORMAT (8HONRES = ,I4/7H VAR = ,G11.5)
      TMP=NRES
      BLN=TMP*VAR*SUMA/SUMB
      WRITE(IOUT,2700) BLN
2700  FORMAT (1H0,5HBN = ,G11.5)
C

```

References Cited

- Bates, D.M., and Watts, D.G., 1980, Relative curvature measures of nonlinearity: *Journal of the Royal Statistical Society, Series B*, v. 42, no. 1, p. 1-25.
- Beale, E.M.L., 1960, Confidence regions in nonlinear estimation: *Journal of the Royal Statistical Society, Series B*, v. 22, no. 1, p. 41-76.
- Guttman, I., and Meeter, D.A., 1965, On Beale's measure of nonlinearity: *Technometrics*, v. 7, no. 4, p. 623-637.
- Linssen, H.N., 1975, Nonlinearity measures: a case study: *Statistica Neerlandica*, v. 29, p. 93-99.
- Theil, H., 1963, On the use of incomplete prior information in regression analysis: *American Statistical Association Journal*, v. 58, no. 302, p. 401-414.
- Vecchia, A.V., and Cooley, R.L., 1987, Simultaneous confidence and prediction intervals for nonlinear regression models with application to a groundwater flow model: *Water Resources Research*, v. 23, no. 7, p. 1237-1250.

Additional Reading

- Cooley, R.L., 1977, A method of estimating parameters and assessing reliability for models of steady-state groundwater flow, 1—Theory and numerical properties: *Water Resources Research*, v. 13, no. 2, p. 318-324.
- _____, 1982, Incorporation of prior information on parameters into nonlinear regression groundwater flow models, 1—Theory: *Water Resources Research*, v. 18, no. 4, p. 965-976.